
2025.06.06 松浦知也（matsura.tomoya@noc.geidai.ac.jp teach@matsuuratomoya.com）

コードとデザイン
東京藝術大学 芸術情報センター開設科目 金曜4-5限 第9週

http://matsura.tomoya@noc.geidai.ac.jp
mailto:teach@matsuuratomoya.com


Arduino Uno R4で音を出す



1.矩形波(1ビット)の波だけで 
いい場合：PWMとtone()



※簡易的な接続のため、直流成分が常にGNDに流れっぱなしになる。

FETが結構発熱するので注意。でかいスピーカーではやらない方がいい

タクトスイッチ



若干膨らんでいて、押すとペコっと凹む



const int button_pin = 2; 
const int sound_pin = 9; 

void setup() { 
  pinMode(button_pin, INPUT_PULLUP); 

  pinMode(sound_pin, OUTPUT); 
} 

void loop() { 
  if (digitalRead(2) == LOW) { 
    tone(sound_pin, 1000); 
  } else { 
    noTone(sound_pin); 
  } 
  delay(20); 
}

tone_minimal_R4.ino



https://akizukidenshi.com/catalog/g/gK-08217/　

ちゃんとした接続の場合はオーディオアンプICを使うのがおすすめ。1個300円ぐらい

https://akizukidenshi.com/catalog/g/gK-08217/


https://ja.aliexpress.com/item/1005007279115127.html

まとめ買いすると1個30円くらいのもある（ただし粗悪品もあるので自己責任）

https://ja.aliexpress.com/item/1005007279115127.html


音名と周波数

• 周波数のままでは音楽用に使いづらい


• 中心のラ（A4）＝440Hz　が一般的なチューニング


• 1オクターブ＝周波数が2倍(A5=880Hz、A6=1760Hz)


• 半音上がるのを12回繰り返すと周波数が2倍になる(＝半音で ずつ上昇)


• A4を69番として、音名に番号を振る（MIDIノート番号）

21/12

Frequency(Hz) = 440 ⋅ 2(midi−69/12)



const int button_pin = 2; 
const int sound_pin = 9; 

bool is_playing = false; 
int button_prev= HIGH; 
float freq = 440; 

float midiToFreq(int midi){ 
  return 440.*pow(2.,(midi-69.)/12.); 
} 

void setup() { 
  pinMode(button_pin,INPUT_PULLUP); 
  pinMode(sound_pin, OUTPUT); 
} 

void loop() { 
  auto button_state = digitalRead(button_pin); 

  if(button_prev == HIGH && button_state==LOW){ 
    is_playing = !is_playing; 
    freq = midiToFreq(random(65,100)); 
  } 
  if(is_playing){ 
    tone(sound_pin,freq); 
  }else{ 
    noTone(sound_pin); 
  } 

  delay(20); 
  button_prev = button_state; 
}

tone_random_flip_R4.ino

Frequency(Hz) = 440 ⋅ 2(midi−69/12)



Mozziで高度な音作り



Mozzi

https://sensorium.github.io/Mozzi/ 
Arduinoで最もポピュラーな音声合成ライブラリ

https://sensorium.github.io/Mozzi/


ライブラリマネージャーでMozziを検索→インストール



※今回は昨年リリースされたv2.0.2を使用します。 
ネットに落ちている資料はまだほぼv1系ばかりで、書き方が変わっている部分もあるので要注意。 

サンプルスケッチを参考にするのが安牌



Arduino Uno R4はA0のピンにDAC(Digital-to-Analog Converter)がついている



※簡易的な接続のため、直流成分が常にGNDに流れっぱなしになる。

FETが結構発熱するので注意。でかいスピーカーではやらない方がいい

1~5Vの範囲で  
音声波形を送る


(0V近辺は波形が歪む)


今日使用する接続方法



Pulse Code Modulation(PCM)

Aquegg~commonswiki, CC-BY-SA 3.0 , https://commons.wikimedia.org/wiki/File:Pcm.svg 

1サンプルごとのビット 
（この例では4ビット）を 
アナログ電圧に直す変換器

(DAコンバーター)が必要 

R4ではA0のピンに12bitの
DACが内蔵

https://commons.wikimedia.org/wiki/File:Pcm.svg


Pulse Density Modulation
縦軸が粗い代わりに横軸を細かくする

Moxfyre, CC 表示-継承 3.0, https://commons.wikimedia.org/w/index.php?curid=3239064 による

ノイズは含まれているが、
高周波に偏っている。 
コンデンサーで高周波を 
取り除くだけで 
そのまま使える。 

（ノイズ成分が20kHz以上
で聞こえなければ 
それすら不要）

Uno R3などでのMozziはPWMピンを使用して擬似的なアナログ波形を実現している

https://commons.wikimedia.org/w/index.php?curid=3239064


Mozzi x Uno R4使用の注意点

• スピーカーへ出力するピンはA0で固定（内蔵DACを使用するため）


• PWMを使った出力には現在非対応（なので、モノラル限定）


• センサーを使う時は、analogRead()の代わりにmozziAnalogRead()を使う必要
がある


• なので、内部的にanalogRead()を呼ぶライブラリとの組み合わせも難しい


• delay()を使うコードはEventDelayクラスを使う必要がある



Mozziの使い方

• setup()とloop()に加えて、void updateControl()とint updateAudio()という関数を作
る


• loop()の中ではaudioHook()という関数だけを呼ぶ 

• 普通loop()に書くような処理（センサーの読み取りなど）はupdateControl()に


• updateAudio()でreturnする値が1サンプルごとの電圧（音圧）になる


• MonoOutput::from8Bit(val)なら0~255、MonoOutput::from16Bit(val)なら
0~32767といった範囲



#define MOZZI_CONTROL_RATE 64   
#include <Mozzi.h> 
#include <Oscil.h> // oscillator template 
#include <tables/sin2048_int8.h> // sine table for oscillator 

Oscil <SIN2048_NUM_CELLS, MOZZI_AUDIO_RATE> aSin(SIN2048_DATA); 

void setup(){ 
  startMozzi();  
  aSin.setFreq(440);  
} 
void updateControl(){ 

} 

AudioOutput updateAudio(){ 
  int16_t output = map((int16_t)aSin.next(),0,255,2000,INT16_MAX); 
  return MonoOutput::from16Bit(output); 
} 

void loop(){ 
  audioHook();  
}

SineWave_R4.ino



#define MOZZI_CONTROL_RATE 64   
#include <Mozzi.h> 
#include <Oscil.h> // oscillator template 
#include <tables/sin2048_int8.h> // sine table for oscillator 

Oscil <SIN2048_NUM_CELLS, MOZZI_AUDIO_RATE> aSin(SIN2048_DATA); 

void setup(){ 
  startMozzi();  
  aSin.setFreq(440);  
} 
void updateControl(){ 

} 

AudioOutput updateAudio(){ 
  int16_t output = map((int16_t)aSin.next(),0,255,2000,32767); 
  return MonoOutput::from16Bit(output); 
} 

void loop(){ 
  audioHook();  
}

SineWave_R4.ino

aSinの波形出力0~255を、 
2000~32767にマッピング 

(0~2000近辺は歪むので使わ
ない)



エンベロープを理解しよう

鍵盤押す 鍵盤離す



エンベロープを理解しよう

Attack Decay

Sustain

Release

Mozziの場合整数で0~255



#define MOZZI_CONTROL_RATE 64  
#include <Mozzi.h> 
#include <Oscil.h>   
#include <mozzi_midi.h> 
#include <ADSR.h> 
#include <tables/sin2048_int8.h> 

Oscil <SIN2048_NUM_CELLS, MOZZI_AUDIO_RATE> 
aOscil(SIN2048_DATA); 

//エンベロープをかけるためのクラス 
ADSR<MOZZI_AUDIO_RATE, MOZZI_AUDIO_RATE> envelope;         

unsigned int Dur, Atk, Dec, Sus, Rel;  
int button_prev = LOW; 

void setup() { 
  pinMode(2,INPUT_PULLUP); 
  startMozzi();  
  Atk = 10; 
  Dec = 10; 
  Sus = 50; 
  Rel = 1000; 
  envelope.setTimes(Atk, Dec, Sus, Rel); 
  envelope.setADLevels(255, 128); 
} 

void updateControl() { 
  auto button_state = digitalRead(2); 
  if (button_prev == HIGH && button_state == LOW) { 
    auto f = mtof((int)random(65, 100)); 
    aOscil.setFreq(f); 
    envelope.noteOn(); 
  } 
  if (button_prev == LOW && button_state == HIGH) { 
    envelope.noteOff(); 
  } 
  button_prev = button_state; 
} 

AudioOutput updateAudio() { 
  envelope.update(); 
  int16_t output_raw = envelope.next() * 
aOscil.next() ; 
  int16_t output = 
map(output_raw,0,65536,5000,32767); 
  return MonoOutput::from16Bit(output); 
} 

void loop() { 
  audioHook(); 
}

mozzi_envelope.ino



#define MOZZI_CONTROL_RATE 64  
#include <Mozzi.h> 
#include <Oscil.h>   
#include <mozzi_midi.h> 
#include <ADSR.h> 
#include <tables/sin2048_int8.h> 

Oscil <SIN2048_NUM_CELLS, MOZZI_AUDIO_RATE> 
aOscil(SIN2048_DATA); 

//エンベロープをかけるためのクラス 
ADSR<MOZZI_AUDIO_RATE, MOZZI_AUDIO_RATE> envelope;         

unsigned int Dur, Atk, Dec, Sus, Rel;  
int button_prev = LOW; 

void setup() { 
  pinMode(2,INPUT_PULLUP); 
  startMozzi();  
  Atk = 10; 
  Dec = 10; 
  Sus = 50; 
  Rel = 1000; 
  envelope.setTimes(Atk, Dec, Sus, Rel); 
  envelope.setADLevels(255, 128); 
} 

void updateControl() { 
  auto button_state = digitalRead(2); 
  if (button_prev == HIGH && button_state == LOW) { 
    auto f = mtof((int)random(65, 100)); 
    aOscil.setFreq(f); 
    envelope.noteOn(); 
  } 
  if (button_prev == LOW && button_state == HIGH) { 
    envelope.noteOff(); 
  } 
  button_prev = button_state; 
} 

AudioOutput updateAudio() { 
  envelope.update(); 
  int16_t output_raw = envelope.next() * 
aOscil.next() ; 
  int16_t output = 
map(output_raw,0,65536,5000,32767); 
  return MonoOutput::from16Bit(output); 
} 

void loop() { 
  audioHook(); 
}

mozzi_envelope.ino

mozzi_midi.hを読み込むと
midiToFreq()相当の関数を

mtof()で使える

値の範囲が 
（0~255）*（0~255）＝

0~65535　になる



http://sheep-me.me/2019/10/24/geidai_16/ 
ひつじさんのサンプルより、Steve Reich「Piano Phase」の再現

http://sheep-me.me/2019/10/24/geidai_16/


#include <Mozzi.h> 
#include <mozzi_midi.h> 
#include <Oscil.h> 
#include <EventDelay.h> 
#include <ADSR.h> 
#include <tables/sin8192_int8.h> 

Oscil<8192, MOZZI_AUDIO_RATE> aOscil(SIN8192_DATA); 
Oscil<8192, MOZZI_AUDIO_RATE> aOscil2(SIN8192_DATA); 
ADSR<MOZZI_AUDIO_RATE, MOZZI_AUDIO_RATE> envelope_A; 
ADSR<MOZZI_AUDIO_RATE, MOZZI_AUDIO_RATE> envelope_B; 
unsigned int Dur, Atk, Dec, Sus, Rel; 

//piano phaseのパターン 
unsigned int pattern[] = { 64, 66, 71, 73, 74, 66, 
64, 73, 71, 66, 74, 73 }; 

//二つのタイマーを用意 
int phase_A = 0; 
int phase_B = 0; 
EventDelay timer_B; 
EventDelay timer_A; 

void setup() { 
  startMozzi(1024); 
  Atk = 10; 
  Dec = 10; 
  Sus = 50; 
  Rel = 100; 
  envelope_A.setTimes(Atk, Dec, Sus, Rel); 
  envelope_A.setADLevels(255, 128); 
  envelope_B.setTimes(Atk, Dec, Sus, Rel); 
  envelope_B.setADLevels(255, 128); 

  timer_A.set(150); 
  timer_B.set(151); 
  timer_B.start(); 
  timer_A.start(); 
} 

void updateControl() { 
  if (timer_A.ready()) { 
    int note = pattern[phase_A]; 
    phase_A = (phase_A + 1) % 12; 
    aOscil.setFreq(mtof(note)); 
    envelope_A.noteOn(); 
    timer_A.start(); 
  } 

  if (timer_B.ready()) { 
    int note = pattern[phase_B]; 
    phase_B = (phase_B + 1) % 12; 
    aOscil2.setFreq(mtof(note)); 
    envelope_B.noteOn(); 
    timer_B.start(); 
  } 
} 

AudioOutput updateAudio() { 
  envelope_A.update(); 
  envelope_B.update(); 
  int16_t A = (envelope_A.next() * aOscil.next()); 
  int16_t B = (envelope_B.next() * aOscil2.next()); 
  int16_t output = map((A + B) / 2, 0, 32767, 2000, 
32767); 
  return MonoOutput::from16Bit(output); 
} 

void loop() { 
  audioHook(); 
} 

mozzi2_reich_R4.ino



Mozziの使い方2

• 今回は使わなかったが、analogReadを使うコードの場合mozziAnalogRead()
に置き換える必要あり


• R4のanalogReadの結果はデフォルトで他のボードと同じ0-1024（10bit）の
範囲になっているが、実はハードウェア的には14bitの精度がある


• analogReadResolution(14)とすると14bitで取得できる


• mozziAnalogRead()は標準で14bit（0-16383）の値を返すので注意

https://sensorium.github.io/Mozzi/doc/html/group__analog.html#gadedb573129313bad0c183af2486c5283


より高度なことがやりたい人は



Daisy

https://www.electro-smith.com/daisy 

Arduino IDE使用可 高音質ステレオDACあり 拡張ボード多数 国内だとFive Gとかで買える

https://www.electro-smith.com/daisy


Bela

https://bela.io/
Linuxが動く。専用のWeb IDEで開発可能

https://bela.io/


参考資料

※Mozziのサンプルはいずれもv1の情報なので参考までに



http://sheep-me.me/講義資料/　
電子工作創作表現 / 音楽音響創造特殊研究 講義資料(2019),ひつじ

http://sheep-me.me/%e8%ac%9b%e7%be%a9%e8%b3%87%e6%96%99/


MozziWS, 中西宣人
https://yoshihito-nakanishi.com/workshops/mozzi 

https://yoshihito-nakanishi.com/workshops/mozzi


https://yoppa.org/ssaw14/ 
SSAW14 – サウンド&ソフトウェアアート 2014, 田所淳

https://yoppa.org/ssaw14/


https://speakerdeck.com/tomoyanonymous/a-bit-high-layer-audio-embedded-platforms-in-2020 

https://speakerdeck.com/tomoyanonymous/a-bit-high-layer-audio-embedded-platforms-in-2020

